CHAPTER 1

LOOKUP FUNCTIONS

InFocus

Excel provides a number of functions that allow you to look up and extract data from a list or table. These are known as *Lookup* functions and they can be used for a variety of purposes, such as:

- returning the appropriate tax rate based on salary
- returning the data that is at, say, the second column, third row of a table
- returning the description, price and discount rate of an item, based on its code in the data inventory

In this session you will:

- ✓ gain an understanding of data lookup functions
- ✓ learn how to use **CHOOSE**
- ✓ learn how to use **VLOOKUP**
- ✓ learn how to use **VLOOKUP** for exact matches
- ✓ learn how to use *HLOOKUP*
- ✓ learn how to use the *INDEX* function
- ✓ learn how to use MATCH
- ✓ gain an understanding of reference functions
- ✓ learn how to use **ROW** and **ROWS**
- ✓ learn how to use **COLUMN** and **COLUMNS**
- ✓ learn how to use ADDRESS
- ✓ learn how to use INDIRECT
- ✓ learn how to use **OFFSET**.

UNDERSTANDING DATA LOOKUP FUNCTIONS

Data lookup functions are used to retrieve data from a table. They generally require at least two pieces of information; **what** to look for and **where** to look for it. The **what** to look for part is often

part of a table of information which can be referred to as a *calculation area*. The *where to look for it* is known as a *data table* – a table in which a list of rates, figures, text or other items are held.

1 Data Area

The *data area* is often on a worksheet by itself, protecting it from accidentally being modified or deleted. It holds all of the possible values for the data. The values are laid out in a table format and they are listed in numerical or alphabetical order of the code that the lookup function will search for. In this example, we have created the name *Pay_Rates* for the range *B3:C7* that holds the data. The resulting formula in the calculation area will be easier to understand.

The name Pay-Rates has been created as a quick way to reference the data table in the range B3:C7...

ᆮ	Pay_Rates	→ (a)	<i>f</i> _x 1	
	Α	В	С	D
1		Hour	ly Rates	
2				
3		1	23.5	
4		2	30.0	
5		3	35.0	
6		4	38.5	
7		5	42.5	
8			Ì	
9				

2 Calculation Area

The *calculation area* is usually on a worksheet by itself unless you require the data values to be visible as well as the resulting calculations.

The calculation area uses a formula, such as VLOOKUP, to find the correct data for each situation. In this example, the VLOOKUP function shown is comparing the value in *C5* with the values in the range *Pay_Rates*. It then returns the value in the second column of the data table, determined by the *2* in the formula.

	D5 ▼		f _x =VLO	OKUP(C5,Pay_	Rates,2)	
4	Α	В	С	D	Е	F
1	Weekly Payroll					
2						
3						
4	First Name	Last Name	Pay Scale	Hourly Rate		
5	Michelle	Calahan	2	\$30.00		
6	Kira	Convery	3			
7	Paddy	Deegan	4			
8	Marty	Doyle	3			
9	Connor	Healy	2			
10	Alana	Keane	1			
11	Siobhan	Kelliher	1			
12	Anthony	O'Brien	3			
13	Melissa	Quinn	4			
14						
15						

The formula here takes the Pay Scale value in C5 (i.e. 2) and finds the corresponding row in the Pay_Rates table (i.e. B4). It then returns the value in the second column of the corresponding row of the Pay-Rates table, which in this case is \$30.00...

USING CHOOSE

The **CHOOSE** function is designed to allow you to make a choice from a list of items. It takes the format of **CHOOSE**(lookup-value, list of items) where the lookup-value indicates the position in

the list of the item you want, that is, 1 being the first item in the list and so on. Range names are often used for the list items to make the function easier to read and comprehend.

Try This Yourself:

Before starting this exercise you MUST open the file E831 Lookup Functions_1.xlsx...

Click on the drop-arrow of the **Name Box** and click on **Rate_1**

The hourly payroll rates have been placed on the Payroll Tables worksheet and each rate has been given a separate name...

- Click on **C4** to see that the second rate has been named **Rate 2**
- Click on the *Payroll* worksheet tab, then click on *D5*

We'll use CHOOSE to look up the pay rate...

- Type =CHOOSE(C5,Rate_1, Rate_2,Rate_3,Rate_4,Rate_5)
- 5 Press Enter
- Click on *D5*, then copy the formula down the list by double-clicking on the fill handle in the bottom right-hand corner of the cell

The hourly rates and the resulting calculated gross pay are now complete

Rate_1		- (e)	f _x	23.5
A	Α	В	С	D
1		Hour	ly Rates	
2				
3		1	2	3.5
4		2	3	0.0
5		3	3.	5.0
6		4	3	8.5
7		5	4	2.5
8				
9				

	D5 ▼	(f _{sc} =CHO	OSE(C5,Rate_	1,Rate_2,Ra	te_3,Rate_4,R	ate_5)	
	Α	В	С	D	Е	F	G	
1	1 Weekly Payroll							
2								
3								
4	First Name	Last Name	Pay Scale	Hourly Rate	Hours Worked	Gross Pay	Tax Rate	
5	Michelle	Calahan	2	\$30.00	12.5	\$375.00		
6	Kira	Convery	3	\$35.00	9.0	\$315.00		
7	Paddy	Deegan	4	\$38.50	16.0	\$616.00		
8	Marty	Doyle	3	\$35.00	35.5	\$1,242.50		
9	Connor	Healy	2	\$30.00	5.0	\$150.00		
10	Alana	Keane	1	\$23.50	41.0	\$963.50		
11	Siobhan	Kelliher	1	\$23.50	2.0	\$47.00		
12	Anthony	O'Brien	3	\$35.00	25.0	\$875.00		
13	Melissa	Quinn	4	\$38.50	32.0	\$1,232.00		
14					-			
15					Totals	\$5,816.00		
16								



1

Note: As you type the formula, you will see the range names appear in a list below the formula. You can click on a name in the list and press Tab to insert it, rather than typing each one...

For Your Reference...

CHOOSE(index_num,value1,value2,...)

This function selects a number, cell reference, defined name, formula, function, or text from a list of up to 254 different options. The *index_num* indicates which of the *values* to select, based on their position in the list.

Handy to Know...

 When you set up a data area for the CHOOSE function, the labels or values to the left of the data play no particular role other than to describe the values listed. This is in contrast to other lookup functions which search the first column of a data table for a match.

USING VLOOKUP

The classic lookup function is **VLOOKUP** – the **vertical lookup**. This function searches vertically down a sorted data table looking for a match with the lookup-value (or the next lowest value). It

then looks across the table to the column you have specified to find the value to return. This is ideal for looking up numeric values within a range, such as tax rates, or commission amounts.

Try This Yourself:

- Continue using the previous file with this exercise, or open the file E831 Lookup Functions_2.xlsx...
- Click on the drop arrow for the **Name Box** and select **Tax_Table**

This is the data table for our VLOOKUP function...

- Click on the *Payroll* worksheet tab to return to the payroll table and click on *G5*
- Type =VLOOKUP(F5,Tax_Table,2)
- ⚠ Press Enter

0% appears because the Gross Pay is less than \$500...

- Copy the formula in *G5* to the range *G6:G13*
- Enter the formula =**F5*G5** in **H5**, then copy it to **H6:H13**
- 7 Enter the formula =F5-H5 in *I5*, then copy it to *I6:I13*
- Click on C10
- O Type 2 and press Enter

The change in the Pay Scale results in changes to the Hourly Rate, Gross Pay, Tax Rate, Tax and Net Pay for Alana Keane

В	С	D	Е	F	G	Н
Hour	Hourly Rates			Tax Ta	ble	
1	23.5			Salary Range	Tax Rate	
2	30.0			\$0.00	0%	
3	35.0			\$500.00	10%	
4	38.5			\$1,000.00	12%	
5	42.5			\$1,200.00	16%	
				\$1,400.00	18%	
				\$1,600.00	20%	
				\$1,800.00	22%	
				\$2,000.00	24%	
				\$2,200.00	26%	
				\$2,400.00	28%	
				\$2,600.00	30%	

ale	Hourly Rate	Hours Worked	Gross Pay	Tax Rate	Тах	Net Pay
	\$30.00	12.5	\$375.00	0%		
	\$35.00	9.0	\$315.00			
	\$38.50	16.0	\$616.00			
	\$35.00	35.5	\$1,242.50			
	\$30.00	5.0	\$150.00			
	\$23.50	41.0	\$963.50			
	\$23.50	2.0	\$47.00			
	\$35.00	25.0	\$875.00			
	\$38.50	32.0	\$1,232.00			

ale	Hourly Rate	Hours Worked	Gross Pay	Tax Rate	Tax	Net Pay
	\$30.00	12.5	\$375.00	0%	\$0.00	\$375.00
	\$35.00	9.0	\$315.00	0%	\$0.00	\$315.00
	\$38.50	16.0	\$616.00	10%	\$61.60	\$554.40
	\$35.00	35.5	\$1,242.50	16%	\$198.80	\$1,043.70
	\$30.00	5.0	\$150.00	0%	\$0.00	\$150.00
	\$30.00	41.0	\$1,230.00	16%	\$196.80	\$1,033.20
	\$23.50	2.0	\$47.00	0%	\$0.00	\$47.00
Γ.	\$35.00	25.0	\$875.00	10%	\$87.50	\$787.50
	\$38.50	32.0	\$1,232.00	16%	\$197.12	\$1,034.88
		Totals	\$6,082.50		\$741.82	\$5,340.68



1

For Your Reference...

VLOOKUP(lookup_value,table,col_index_num)

This function searches down the left-most column of the *table* until it finds the *lookup-value* or the row with the next lowest value. It then refers to the *column index number* nominated in the function, and returns the value found in the corresponding row and column.

Handy to Know...

 An alternative to the VLOOKUP function is the horizontal or HLOOKUP function. This looks for a value in the top row of a table or array of values, and returns the value from the same column in the row you specify.

USING VLOOKUP FOR EXACT MATCHES

Generally, the **VLOOKUP** function uses three arguments: the lookup-value, the table location and the column number. This lookup will find a row based on a numeric range. However, you

may only want to return a value if you get an exact match, and the optional fourth argument, *match-type*, makes this possible. By adding *FALSE* to the function, *#NA* is returned if a match isn't found.

Try This Yourself:

Continue using the previous file with this exercise, or open the file E831 Lookup Functions_3.xlsx...

Click on the drop arrow for **Name Box** and click on **Items List**

This lookup table includes specific product codes that require an exact match...

Click on the *Invoice* worksheet tab, then click on *C7*

We'll use the exact lookup to find the description...

- Type =VLOOKUP (B7,ltems_List,2,FALSE)
- Click on *E7*, type =VLOOKUP(B7,Items_List,3, FALSE), then press Enter

We only want to charge a deposit if B3 is not blank...

- Click on F7, type =IF(ISBLANK(\$B\$3),0, VLOOKUP(B7,Items_List,4, FALSE)), then press [Enter]
- Click on **G7**, type =(D7*E7)*(1-F7), then press Enter
- Copy the formulas in columns

 C, E, F and G down to Row 10

 to complete the invoice as

 shown

Item	Description	Quantity	Price	Deposit	Total Cost
TEL00001	=VLOOKUP(B7,Items_List,2,FA	LSE)			
TEL00003		1			
TEL00005		2			
TEL00007		1			



Item	Description	Quantity	Price	Deposit	Total Cost
TEL00001	World Communicator 223	2	\$234.50	12%	\$412.72
TEL00003		1			
TEL00005		2		Ĭ	,
TEL00007		1			



Item	Description	Quantity	Price	Deposit	Total Cost	
TEL00001	World Communicator 223	2	\$234.50	12%	\$412.72	
TEL00003	Master Communicator 10 Plus	1	\$1,237.90	22%	\$965.56	
TEL00005	Global Roamer 514	2	\$237.80	22%	\$370.97	
TEL00007	Global Roamer 516	1	\$677.00	12%	\$595.76	Į
			To	tal Items	\$2,859.50	
			Total	Deposit	\$514.49	
			Tota	l Invoice	\$3,373.99	



For Your Reference...

To use VLOOKUP for exact matching:

VLOOKUP(lookup_value,table,col,range_lkup)

Use the value of **FALSE** for **range_Ikup** when you want to ensure exact matches only. **#NA** will be returned if an exact match isn't found.

Handy to Know...

 You can use the ISNA(value) function to trap #NA results. For example, the structure =IF(ISNA(lookup), "Code not found",lookup) will display a useful message when an exact match is not possible, rather than #NA which does not explain the problem clearly.

USING HLOOKUP

Another commonly used lookup function is **HLOOKUP** – the **horizontal lookup**. This function searches horizontally across a sorted data table looking for a match with the lookupvalue (or the next lowest value). It then looks down the table to the row you have specified to find the value to return. This can be used to search for text, numbers, or logical values.

Try This Yourself:

Same File Continue using the previous file with this exercise, or open the file E831 Lookup Functions_4.xlsx...

Click on the *Index* worksheet tab then click on *D14* and examine the formula

This is an example of the VLOOKUP function. This scans down the list of options looking for the corresponding Equipment number. When it finds a match, it returns the description.

You can use HLOOKUP in a similar way...

- Click on cell *D15* and type =HLOOKUP(C15,Rate_Type, 2,TRUE)
- ? Press Enter

The name of the Rate Type will be returned and displayed...

Experiment with different values in cells **C14** and **C15**

You'll find that the formulas return the corresponding descriptions

		014 ▼ (f _x =VLC	OKUP(C14,E	quipment_L	ist,2,FALSE)					
1	Α	В	С	D	Е	F				
1	Communications Equipment Hire Costs									
2										
3			1	2	3	4				
4	No	Description	Corporate	Frequent	Private	Staff				
5	1	World Communicator 223	\$60.00	\$51.00	\$66.00	\$30.00				
6	2	Planet Tamer 34e	\$75.00	\$63.75	\$82.50	\$37.50				
7	3	Master Communicator 10 Plus	\$120.00	\$102.00	\$132.00	\$60.00				
8	4	Global Roamer 514	\$60.00	\$51.00	\$66.00	\$30.00				
9	5	Global Roamer 515	\$75.00	\$63.75	\$82.50	\$37.50				
10	6	Global Roamer 516	\$85.00	\$72.25	\$93.50	\$42.50				
11	7	Global Roamer 517	\$95.00	\$80.75	\$104.50	\$47.50				
12										
13				Description						
14		Equipment Item No	5	Global Roan	ner 515					
15		Rate Type	1							
16		Number of Days	10							
17										



1		Communication	ıs Equipr	nent Hire	e Costs	
2						
3			1	2	3	4
4	No	Description	Corporate	Frequent	Private	Staff
5	1	World Communicator 223	\$60.00	\$51.00	\$66.00	\$30.00
6	2	Planet Tamer 34e	\$75.00	\$63.75	\$82.50	\$37.50
7	3	Master Communicator 10 Plus	\$120.00	\$102.00	\$132.00	\$60.00
8	4	Global Roamer 514	\$60.00	\$51.00	\$66.00	\$30.00
9	5	Global Roamer 515	\$75.00	\$63.75	\$82.50	\$37.50
10	6	Global Roamer 516	\$85.00	\$72.25	\$93.50	\$42.50
11	7	Global Roamer 517	\$95.00	\$80.75	\$104.50	\$47.50
12						
13				Description		
14		Equipment Item No	5	Global Roan	ner 515	
15		Rate Type	1	Corporate		
16		Number of Days	10			
17						



For Your Reference...

HLOOKUP(lookup_value,table,row_index_num)

This function searches across the top-most column of the *table* until it finds the *lookup_value* or the row with the next lowest value. It then refers to the *row index number* nominated in the function, and returns the value found in the same column and corresponding row.

Handy to Know...

- If the *lookup_value* is smaller than the smallest value in the first row of *table*, HLOOKUP will return the *#N/A* error value.
- Range_lookup is an optional argument. If TRUE or omitted, it will find the closest match in the top row. FALSE forces an exact match.

USING INDEX

The *INDEX* function is used to look up values according to specific row and column locations. The function requires the *location* of the lookup table, and the *row* and *column* to look in. This is

especially useful for two-dimensional tables where you want the lookup values to be visible. For example, you may want alternative prices to be visible with your price calculator.

Try This Yourself:

Same File Continue using the previous file with this exercise, or open the file E831 Lookup Functions_5.xlsx...

Click on the drop arrow for the *Name Box* and click on *Rate_List*

This will take you to the lookup table which shows the various hire rates for communications equipment...

- Click on C18 and type =INDEX(Rate_List,C14,C15)
- ? Press Enter

This will vary depending upon which figures appear in C14 and C15...

Click on *C19*, type =C16*C18, then press Enter

The function has looked up the figure for the equipment type for the rate, and has returned the value. The total multiplies the number of days hire by the hire rate to give the Total Hire cost...

Click on *C14*, type 1, then press Enter

The value returned is that for equipment type 1 with the given rate type



Description	Corporate	Frequent	Private	Staff
World Communicator 223	\$60.00	\$51.00	\$66.00	\$30.00
Planet Tamer 34e	\$75.00	\$63.75	\$82.50	\$37.50
Master Communicator 10 Plus	\$120.00	\$102.00	\$132.00	\$60.00
Global Roamer 514	\$60.00	\$51.00	\$66.00	\$30.00
Global Roamer 515	\$75.00	\$63.75	\$82.50	\$37.50
Global Roamer 516	\$85.00	\$72.25	\$93.50	\$42.50
Global Roamer 517	\$95.00	\$80.75	\$104.50	\$47.50
		Description		
Equipment Item No	2	Planet Tame	er 34e	
Rate Type	4	Staff		
Number of Days	10			
=INDEX(Rate_List	,C14,C15)]		
Total Hire				



Description	Corporate	Frequent	Private	Staff
World Communicator 223	\$60.00	\$51.00	\$66.00	\$30.00
Planet Tamer 34e	\$75.00	\$63.75	\$82.50	\$37.50
Master Communicator 10 Plus	\$120.00	\$102.00	\$132.00	\$60.00
Global Roamer 514	\$60.00	\$51.00	\$66.00	\$30.00
Global Roamer 515	\$75.00	\$63.75	\$82.50	\$37.50
Global Roamer 516	\$85.00	\$72.25	\$93.50	\$42.50
Global Roamer 517	\$95.00	\$80.75	\$104.50	\$47.50
Fauinment Item No.	2	Description Planet Tamer	240	
Equipment Item No			346	
Rate Type	-	Staff		
Number of Days	10			
Hire Per Day	\$37.50			
Total Hire	\$375.00			



		Description
Equipment Item No	1	World Communicator 223
Rate Type	4	Staff
Number of Days	10)
Hire Per Day	\$30.00	
Total Hire	\$300.00	

For Your Reference...

INDEX(array,row,column)

This function goes to the table at the *array* or range specified and returns the value found in the specific *row* and *column* of the table.

Handy to Know...

 The INDEX function can also be used for multiple tables, where it takes the form:

INDEX(reference,row,column,area)

 Here reference refers to one or more ranges, and area refers to which of these ranges to look in for the value.

Using MATCH

The **MATCH** function looks for an item in a list and returns the position of that item. **MATCH** is used instead of the other lookup functions when the *position* of the item rather than the item itself is needed. For example, you can search for a particular category name in an alphabetical list and return its location in the list, such as 3 for the 3" position or 4 for the 4th position.

Try This Yourself:

Continue using the previous file with this exercise, or open the file E831 Lookup Functions_6.xlsx...

On the Index worksheet, click on C18 and examine the formula

> At the moment, the formula uses the Rate Type in C15 to determine which column to look in. Using MATCH you can allow the user to type the rate name rather than a number...

- Click on cell C15 and type Staff, then press Tab and delete the contents of D15 which will display an error
- Click on C18, then double-click on C15 in the formula to select it - we'll replace it with the MATCH function
- Type MATCH(C15,C4:F4)
- Press Enter

The formula looks up the text and returns its position, which is 4 in this case. This is, in turn, used by the INDEX function to locate the correct hire rate for the given rate type





	MIN ▼ (* X ✔ fx = INDEX(Rate_List,C14,MATCH(C15,C4:F4))									
4	Α	B INDEX(array, row_num, [column_num])								
1	1 Communications Equipment Hire Costs									
2										
3			1	2	3	4				
4	No	Description	Corporate	Frequent	Private	Staff				
5	1	World Communicator 223	\$60.00	\$51.00	\$66.00	\$30.00				
6	2	Planet Tamer 34e	\$75.00	\$63.75	\$82.50	\$37.50				
7	3	Master Communicator 10 Plus	\$120.00	\$102.00	\$132.00	\$60.00				
8	4	Global Roamer 514	\$60.00	\$51.00	\$66.00	\$30.00				
9	5	Global Roamer 515	\$75.00	\$63.75	\$82.50	\$37.50				
10	6	Global Roamer 516	\$85.00	\$72.25	\$93.50	\$42.50				
11	7	Global Roamer 517	\$95.00	\$80.75	\$104.50	\$47.50				
12										



	C	19 ▼ (f _x =C16	5*C18			
1	Α	В	С	D	Е	F
1		Communication	s Equipr	nent Hire	Costs	
2						
3			1	2	3	4
4	No	Description	Corporate	Frequent	Private	Staff
5	1	World Communicator 223	\$60.00	\$51.00	\$66.00	\$30.00
6	2	Planet Tamer 34e	\$75.00	\$63.75	\$82.50	\$37.50
7	3	Master Communicator 10 Plus	\$120.00	\$102.00	\$132.00	\$60.00
8	4	Global Roamer 514	\$60.00	\$51.00	\$66.00	\$30.00
9	5	Global Roamer 515	\$75.00	\$63.75	\$82.50	\$37.50
10	6	Global Roamer 516	\$85.00	\$72.25	\$93.50	\$42.50
11	7	Global Roamer 517	\$95.00	\$80.75	\$104.50	\$47.50
12						
13				Description		
14		Equipment Item No	1	World Comn	nunicator 22	3
15		Rate Type	Staff			
16		Number of Days	10			
17						
18		Hire Per Day	\$30.00			
19		Total Hire	\$300.00			
20						



For Your Reference...

MATCH(lookup_value,table,match_type)

This function searches through the table until it finds the lookup value or the row with the next lowest value. It then returns the position of the value in the array. *Match_type* can be 1 (find value less than or equal to lookup_value), 0 (exact match) or -1 (value greater than or equal to).

Handy to Know...

When **MATCH** searches for a text value, it expects to find the list in alphabetical order. If it doesn't find an exact match in the alphabetical search, it will return the position of the word that would have appeared alphabetically before it in the list. This can be varied by changing match-type.

UNDERSTANDING REFERENCE FUNCTIONS

Reference functions are functions that work with cell, row, column and range references. They are used to locate cells or ranges, or to identify independently the row of a cell or the

column of a cell. Mostly, they are used for quite advanced formulas and so it is important that you have an understanding of how they work and what the term *cell reference* really means.

Cell References

You are probably comfortable referring to a cell by its *cell reference* which is a combination of row and column reference, for example, *B1*. The *B* refers to column *B* and the *1* refers to row *1*. Behind the scene, Excel converts the column reference to the numerical equivalent, so, for example, the contents of cell *B1* would be found in column *2*, row *1*. Some people even like to use the format *R1C2*, where this reference reads as row *1* column *2*.

Reference Functions

Reference functions work with the parts of **cell references**, such as the **row reference**, or with complete **cell references** so that they can locate data or return information about or from cells. Here is a simple example of some reference functions and an explanation in the table below.

4	Α	В	С	D	Е	F
1						
2			Function	Formula Example	Result	
3			ROW	=ROW()	3	
4			COLUMN	=COLUMN()	5	
5			ADDRESS	=ADDRESS(E3,E4)	\$E\$3	
6			INDIRECT	=INDIRECT(E5)	3	
7			OFFSET	=OFFSET(E5,-1,0)	5	
8						
9						

ROW	Returns the row number of a reference. In the example above, <i>ROW()</i> returns the current row number which is 3 .
COLUMN	Returns the column number of a reference. In the example above, <i>COLUMN()</i> returns the current column number which is <i>5</i> .
ADDRESS	Returns a reference as <i>text</i> to a single cell in a worksheet. In the example above, <i>ADDRESS</i> uses the <i>row</i> number in <i>E3</i> (3) and the <i>column</i> number in <i>E4</i> (5) to construct the <i>cell address \$E\$3</i> .
INDIRECT	Returns a reference indicated by a text value. This is often used in conjunction with <i>ADDRESS</i> to return the value from a specific cell reference created by <i>ADDRESS</i> . In the example above, <i>INDIRECT</i> refers to the text address created by <i>ADDRESS</i> in <i>E5</i> , then looks in the cell <i>\$E\$3</i> and returns the contents which is <i>3</i> .
OFFSET	Returns a reference offset from a given reference. In the example above, OFFSET looks at cell E5 , then goes to the cell one above (-1) and in the same column (0) and returns the contents which is 5.
ROWS	Returns the number of rows in a reference.
COLUMNS	Returns the number of columns in a reference.
AREAS	Returns the number of non-contiguous areas in a range. Used for sophisticated programming.
TRANSPOSE	Returns a vertical array of cells as a horizontal array or vice versa. Used for sophisticated programming.

USING ROW AND ROWS

The **ROW** function returns the row number of a reference. The **ROWS** function, on the other hand, returns the number of rows in a reference. You can use **ROW** to create row numbering, or in

conjunction with other functions, such as **ADDRESS**, to construct cell addresses from information located in other parts of a spreadsheet.

Try This Yourself:

- Before starting this exercise you MUST open the file E831 Lookup Functions_7.xlsx...
- On the **Addresses** worksheet, click on **A1**, type **=ROW()** and press Enter
- Click back on *A1* then drag from the fill handle down to *A9*

This copies the formula down and gives you row numbering. What if you insert a row?

Click on the row header for row 5, then click on *Insert*, in the *Cells* group, on the *Home* tab

The numbers automatically adjust...

- Copy the formula in **A4** down to **A5** to complete the numbering
- Click on A12 and type =ROWS(A1:A10), then press

This counts the number of rows in the range that you've specified, whether or not they have any contents

		А	В	С	D	Е	F	
U	1	1						
	2							
	3							
	4							
	5							

		А	В	С	D	Е	F	
2	1	1						
	2	2						
	3	3						
	4	4						
	5	5						
	6	6						
	7	7						
	8	8						
	9	9						
	10		=					
	11							

3		А	В	С	D	Е	F	
3	1	1						
	2	2						
	3	3						
	4	4						
	5							
	6	6						
	7	7						
	8	8						
	9	9						
	10	10						
	11							

	Α	В	С	D	Е	F	
1	1				_	•	
2	2						
3	3						
4	4						
5	5						
6	6						
7	7						
8	8						
9	9						
10	10						
11							
12	10						
13							
1/1							

For Your Reference...

ROW(reference)

This function returns the row number of the reference. The reference is optional.

ROWS(reference)

This function returns the number of rows in the reference.

Handy to Know...

 The ROW and COLUMN functions are often used together to create a complete cell reference.

USING COLUMN AND COLUMNS

The **COLUMN** function returns the column number of a reference. The **COLUMNS** function, on the other hand, returns the number of columns in a reference. **COLUMN** can be used to create

column numbering or used in conjunction with other functions, such as **ADDRESS** and **INDIRECT**, to enable access to specific cells in a worksheet.

Try This Yourself:

Continue using the previous file with this exercise, or open the file E831 Lookup Functions_8.xlsx...

- Click on *B1* and type =*COLUMN()*, then press Enter
- Click on **B1**, then drag the fill handle across to **G1** to copy the formula

Like rows, this returns the column number of the cell in which it resides...

Click on *B12* and type =*COLUMNS(A1:G1)*, then press Enter

Excel counts 7 columns in this range



	1	Α	В	С	D	Е	F	
1	1	1	2					
:	2	2						
:	3	3						
-	4	4						

	Α	В	С	D	Е	F	G	
1	1	2	3	4	5	6	7	
2	2							==
3	3							
4	4							
5	5							
6	6							
7	7							
8	8							
9	9							
10	10							
11								



A	Α	В	С	D	Е	F	G	
1	1	2	3	4	5	6	7	
2	2							
3	3							
4	4							
5	5							
6	6							
7	7							
8	8							
9	9							
10	10							
11								
12	10	7						
13								
14								



For Your Reference...

COLUMN(reference)

This function returns the column number of the reference. The reference is optional.

COLUMNS(reference)

This function returns the number of columns in the reference.

Handy to Know...

 COLUMN() returns the column number rather than the letter that we would usually use to refer to a column. Excel is able to interpret this number via other formulas to work out which column to look in.

USING ADDRESS

The **ADDRESS** function creates a text version of a cell address from a given row number and a given column number. It's often used in conjunction with **ROW** and **COLUMN**, where

1

ROW and **COLUMN** retrieve the two parts of the cell reference separately. **ADDRESS** combines a row reference and a column reference to create a cell reference.

Try This Yourself:

Same

Continue using the previous file with this exercise, or open the file E831 Lookup Functions_9.xlsx...

On the *Addresses* worksheet, click on *B2* and type

=*ADDRESS(\$A2,B\$1,4)*then press Enter and click back on *B2*

This formula creates the cell reference from the row number in A2 and the column number in B1. The partial absolute addresses are used in the formula to allow it to be copied to the rest of the table...


Using the fill handle, drag the formula down to row 10 and across to column **G**

This calculates the cell address of each cell from the row and column addresses in column A and row 1...

3 Click elsewhere in the spreadsheet to see the result

	B2 ▼ (**		f _x	=ADDRESS(\$A2,B\$1,4)						
	Α	В	С	D	Е	F	G	ı		
1	1	2	3	4	5	6	7			
2	2	B2								
3	3									
4	4									
5	5									
6	6									
7	7									
8	8									
9	9									
10	10									
11										
12	10	7								
13										





For Your Reference...

ADDRESS(row_num, col_num)

This function creates a text version of a cell reference from the **row_num** and **col_num**. By default it creates an absolute cell reference, but this can be altered by using the optional parameter **abs_num** where 1=absolute, 2=absolute row, 3= absolute column and 4=relative.

Handy to Know...

ADDRESS has three optional parameters which are abs_num, r1 and sheet_text respectively. If r1 is FALSE, it gives a R1C1 style reference. Otherwise, it is A1.
 Sheet_text is the text to be used as the worksheet name. It appears before the cell reference in the cell.

USING INDIRECT

The **INDIRECT** function returns the reference specified by a text string. Often the text string is provided by the **ADDRESS** function or the **ROW** and **COLUMN** functions. When the reference is

returned, the contents of the referenced cell are displayed. This means that **INDIRECT** is often used to retrieve the contents of a cell located by another function, such as the **MAX** or **MIN**.

Try This Yourself:

Continue using the previous file with this exercise, or open the file E831 Lookup Functions_10.xlsx...

Click on the **INDIRECT**worksheet tab and click on **E3**


This includes property prices as well as the maximum, minimum and median house prices. The list of sales prices is named Sales_Figures...

Click on F3 and type
=ADDRESS(MATCH(
Top_Price,Sales_Figures,0) +
ROW(B2),COLUMN(B2)) then
press Enter and click back on F3

This tells you that the top price is in \$B\$11. We need the property's address which is in the cell one to the left...

- Glick in front of **ADDRESS** in the formula and type **INDIRECT**(
- Press End to move to the end of the formula, then edit as shown
- Press Enter, then in F4 type
 =INDIRECT(ADDRESS(MATCH(
 Min_Price,Sales_Figures,0) +
 ROW(B2),COLUMN(B2)-1)) and
 press Enter


Now the address of the minimum price house is also visible







You need to type -1) in front of the last (closing) bracket



For Your Reference...

INDIRECT(ref text)

Retrieves the contents of the cell with the address *ref_text*.

Handy to Know...

 By default, INDIRECT expects an A1 format cell reference. To use a R1C1 style reference, use the optional parameter a1.
 For example, INDIRECT("B3",TRUE) and INDIRECT("R3C2", FALSE) will both return the contents of B3.

USING OFFSET


The **OFFSET** function is used to retrieve a cell or range that is a specified number of rows and/or columns from a given cell or range of cells. It is great for locating information in a list when you

know where the data will be positioned in relation to other figures with a known location. *OFFSET* can return a single cell or a range of cells.

Try This Yourself:

Continue using the previous file with this exercise, or open the file E831 Lookup Functions_11.xlsx...

- Click on the **OFFSET**worksheet tab and click on **F3**
- Type =OFFSET(INDIRECT(
 ADDRESS(MATCH(
 Max_Read,Readings,0)+
 ROW(B3),COLUMN(B3)))
 ,0,-1)
 then press Enter
- In F4 type
 =OFFSET(INDIRECT(
 ADDRESS(MATCH(
 Min_Read,Readings,0)+
 ROW(B3),COLUMN(B3)))
 ,0,-1)
 then press Enter Enter





4	Α	В	С	D	E	F	G
1	Daily R	Readir	ıgs				
2					Time of First Occurence		
3	Time	Reading		Maximum	50	2:45:00 PM	
4	12:00:00 PM	44.00		Minimum	1		
5	12:15:00 PM	22.00		Average	27.15		
6	12:30:00 PM	27.00					
7	12:45:00 PM	26.00					
8	1:00:00 PM	45.00					



	Α	В	С	D	Е	F	G	
1	Daily Readings							
2						Time of First Occurence		
3	Time	Reading		Maximum	50	2:45:00 PM		
4	12:00:00 PM	44.00		Minimum	1	4:30:00 AM		
5	12:15:00 PM	22.00		Average	27.15			
6	12:30:00 PM	27.00						
7	12:45:00 PM	26.00						
8	1:00:00 PM	45.00						



For Your Reference...

OFFSET(reference, rows, cols)

This function goes to the cell **reference**, then moves down the number of **rows** and across the number of **cols** specified, then retrieves the data in the cell or range it locates.

Handy to Know...

 The additional optional parameters height and width determine how many cells are returned by the OFFSET function. For example, OFFSET(B3,2,1,2,2) will retrieve the range C5:D6. In this case, it is often used with SUM to calculate the total of these cells.